Usando el Control para crear Hojas Electronicas de Carlos Aguilar Mares
El programador mexicano Carlos Aguilar Mares ha creado algunos controles gratis de mucha utilidad para hacer gráficas, etc. Uno de estos controles es el ExcelXMLWriter, para escribir una hoja de Excel, aún sin tener Excel en la máquina.

El sitio de Carlos Aguilar Mares es: http://www.carlosag.net, de donde se pueden bajar los controles que amablemente 
Creando el Proyecto DevCareSheet
Vamos a crear un proyecto. Para que todos los archivos queden en un solo lugar, crearemos un folder en esta dirección: C:\DevCare\DevCareSheet.

Allí copiaremos el módulo de Excel ExcelXMLWriter, con el nombre CarlosAg.ExcelXmlWriter.dll:

Con el administrador del Internet Information Server creamos un Virtual Directory bajo el sitio web por default (Default Web Site). El Virtual Directory se llamará DevCareSheet y apuntará a C:\DevCare\DevCareSheet.

Luego, en Visual Studio .NET creamos el nuevo proyecto DevCareSheet:

Creando las Hojas Web del Sitio
Crearemos dos hojas Web: Una hoja Web, wfSheet.aspx, para que el usuario pueda ingresar algún input – en nuestro caso, preguntará por el país, del cual queremos que baje datos de venta de clientes. La otra hoja Web, wfSheetDownload,aspx, que no contiene nada, y es la que hace el trabajo de bajar la hoja de Excel.
Nota: Por default, la hoja de Excel toma el nombre de la hoja Web que hace el bajado. En nuestro ejemplo, la hoja de Excel se llamaría: wfSheetDownload.XLS. Considere cambiar el nombre de la hoja a algún nombre más apropiado (¿Tal vez VentasClientesXPais.aspx?).

Hoja Web: wfSheet.aspx

El sitio va a comenzar con esta página Web (en el Solution Explorer, sobre la hoja wfSheet.aspx dar click derecho y pedir Set as Start Page). La página le pregunta al usuario por un país, para que baje los clientes y las ventas de sólo el país seleccionado.
En este ejemplo, pusimos la propiedad pageLayout en FlowLayout. (Ud. especifique como le parezca mejor. Sólo recuerde que la otra opción, GridLayout sólo funciona con navegadores que manejen DHTML).

Dibujamos 3 controles. Un label: lblPromptBajarClientesDePais, un text box: txtPais, y un botón: btnBajar.

Código de la Página wfSheet.aspx

Escribimos el siguiente código para el botón btnBajar (dé doble click sobre el botón btnBajar). Lo que va a hacer es redireccionar hacia la página de bajada de Excel: wfSheetDownload.aspx.
    Private Sub btnBajar_Click( _

        ByVal sender As System.Object, _

        ByVal e As System.EventArgs) Handles btnBajar.Click
        Response.Redirect("wfSheetDownload.aspx?country=" & Me.txtPais.Text)
    End Sub
Hoja wfSheetDownload.aspx

Esta hoja no contendrá controles gráficos. Acá sólo reside la lógica de bajada de archivo de Excel.
En mi caso, a la página Web le cambié la propiedad pageLayout a flowLayout, para evitar problemas con navegadores. Recuerde que flowLayout, a diferencia de GridLayout, es más genérico, pues no utiliza el lenguaje DHTML, que sólo algunos navegadores pueden ver.
Referencia al Módulo de Carlos Aguilar Mares
En el Solution Explorer, bajo References, agregue una referencia al control de Carlos Aguilar Mares:

Código Fuente de la Hoja wfSheetDownload.aspx

Pedimos ver el código de la página, y escribimos lo siguiente
Imports CarlosAg.ExcelXmlWriter

Public Class wfSheetDownload

    Inherits System.Web.UI.Page
(Web Form Designer Generated Code)

    Private Sub Page_Load( _

         ByVal sender As System.Object, _

         ByVal e As System.EventArgs) Handles MyBase.Load

        'Put user code to initialize the page here

        '-----------------------------------------------------------------

        ' Lee parámetro country pasado por la página llamante

        ' Request Item = "country"

        '-----------------------------------------------------------------

        Dim pCountry As String

        pCountry = Request.Item("country")

        '-----------------------------------------------------------------

        ' Trae datos para llenar la hoja electronica

        '-----------------------------------------------------------------

        Const TBL_CUSTOMERS = "Customers"

        Const ITEM_COMPANYNAME = "CompanyName"

        Const ITEM_CONTACTNAME = "ContactName"

        Const ITEM_COUNTRY = "Country"

        Const ITEM_CITY = "City"

        Const ITEM_TOTALVALUE = "TotalValue"

        Dim sCnn As String = _

            "server=(local);integrated security=sspi;database=northwind"

        Dim sSQL As String = _

            "SELECT " _

            & " A.CompanyName, " _

            & " A.ContactName, " _

            & " A.Country, " _

            & " A.City, " _

            & " SUM(C.UnitPrice * C.Quantity) as TotalValue " _

            & " FROM customers A " _

            & "   LEFT JOIN Orders B " _

            & "      ON A.CustomerID=B.CustomerID " _

            & "   LEFT JOIN ""Order Details"" C " _

            & "      ON B.OrderID=C.OrderID " _

            & " WHERE country=@country " _

            & " GROUP BY " _

            & " A.CompanyName, A.ContactName, A.Country, A.City "

        Dim ds As New DataSet

        Dim da As New SqlClient.SqlDataAdapter(sSQL, sCnn)

        ' Parámetro @country

        Dim paramCountry As New _

            SqlClient.SqlParameter("@country", SqlDbType.VarChar)

        paramCountry.Value = pCountry

        da.SelectCommand.Parameters.Add(paramCountry)

        da.Fill(ds, TBL_CUSTOMERS)

        '-----------------------------------------------------------------

        ' Ahora creará la hoja electrónica

        '-----------------------------------------------------------------

        ' Cuántas filas dejará arriba antes de loopear

        Const ROWS_DOWN = 4

        ' Libro

        Dim bookClientesPais As New Workbook

        ' Hoja

        Dim sheetClientesPais As Worksheet = _

            bookClientesPais.Worksheets.Add("Clientes x Pais")

        ' Columnas (no necesario... lo puse para controlar ancho)

        sheetClientesPais.Table.Columns.Add(New WorksheetColumn(150))

        sheetClientesPais.Table.Columns.Add(New WorksheetColumn(100))

        ' Crearemos un estilo para los números

        Dim styleNumbers As New WorksheetStyle("StyleNumbers")

        styleNumbers.Alignment.Horizontal = StyleHorizontalAlignment.Right

        styleNumbers.NumberFormat = "#,##0.00"

        styleNumbers.Interior.Color = "Cyan"

        styleNumbers.Interior.Pattern = StyleInteriorPattern.Solid

        bookClientesPais.Styles.Add(styleNumbers)

        ' Loop de llenado de filas y celdas

        Dim i As Int64

        For i = 0 To ds.Tables(TBL_CUSTOMERS).Rows.Count - 1

            ' Variables temporales, para leer datos de 1 cliente

            Dim sCompanyName As String

            Dim sContactName As String

            Dim sCountry As String

            Dim sCity As String

            Dim dTotalValue As Decimal = 0


            With ds.Tables(TBL_CUSTOMERS).Rows(i)

                sCompanyName = .Item(ITEM_COMPANYNAME)

                sContactName = .Item(ITEM_CONTACTNAME)

                sCountry = .Item(ITEM_COUNTRY)

                sCity = .Item(ITEM_CITY)

                dTotalValue = .Item(ITEM_TOTALVALUE)

            End With

            ' Nueva fila

            Dim rowCliente As WorksheetRow = _

                sheetClientesPais.Table.Rows.Add


            ' Celdas de la Fila, e índice de la Fila

            With rowCliente

                .Cells.Add( _

                   New WorksheetCell(sCompanyName, DataType.String))

                .Cells.Add( _

                   New WorksheetCell(sContactName, DataType.String))

                .Cells.Add( _

                   New WorksheetCell(sCountry, DataType.String))

                .Cells.Add( _

                   New WorksheetCell(sCity, DataType.String))

                .Cells.Add( _

                New WorksheetCell( _

                    dTotalValue.ToString, _

                    "StyleNumbers"))

                .Index = ROWS_DOWN + i

            End With

        Next


        ' Lo manda por la Web de regreso al usuario...

        'bookClientesPais.Save("c:\DevCare\DevCareSheet\test.xls")

        Response.ContentType = "application/vnd.ms-excel"

        bookClientesPais.Save(Response.OutputStream)

        'Response.Redirect("wfSheet.aspx")

    End Sub

End Class
Código HTML

Además, en el código HTML de la página wfSheetDownload.aspx, eliminamos todo lo que va bajo la primera directiva, y modificamos esta directiva para que diga lo siguiente – con especial énfasis en el ContentType=”text/html”.
<%@ Page Language=”vb” AutoEventWireup=”false” Codebehind=”wfSheetDownload.aspx.vb” Inherits=”DevCareSheet.wfSheetDownload” ContentType=”text/html” %>

Corriendo el Programa

Al correrlo, nos debería aparecer la página wfSheet.aspx:

Escoja Open para abrir la página inmediatamente, o Save para guardar la página en el disco.

Yo escogí Open y apareció lo siguiente:

Note que todavía NO he logrado que me aparezcan bien los formatos de dinero… Investigar…

Creamos estas dos páginas Web

